

使用 HD44780 (KS0066U) 的字符型液晶显示模块

字符型液晶显示模块由字符型液晶显示屏 (LCD)，控制驱动主电路 HD44780 及其扩展驱动电路，少量阻、容元件，结构件等装配在 PCB 板上而成。

字符型液晶显示模块目前在国际上已经规范化，无论显示屏规格如何变化，其电特性和接口形式都是比较统一的。因此只要设计出一种型号的接口电路，在指令设置上稍加改动即可使用各种规格的字符型液晶显示模块。

● 特点描述

- 1、液晶显示屏是以若干个 5×8 或 5×11 点阵块组成的显示字符群。每个点阵块为一个字符位，字符间距和行距都为一个点的宽度。
- 2、主控制驱动电路为 HD44780(HITACHI) 及其他公司全兼容电路，如 KS0066(SAMSUNG)、SPLC780(SUNPLUS)。
- 3、具有字符发生器 ROM 可显示 192 种字符(160 个 5×7 点阵字符和 32 个 5×10 点阵字符)。
- 4、具有 64 个字节的自定义字符 RAM，可自定义 8 个 5×8 点阵字符或四个 5×11 点阵字符。
- 5、具有 80 个字节的 RAM。
- 6、标准的接口特性，适配 M6800 系列 MPU 的操作时序。

● 基本特性

一、电特性

(1) 极限参数

项目	符号	最小值	最大值	备注
电路逻辑电压	Vdd-Vss	0	7.0V	
液晶驱动电压	Vdd-Vo	0	13.5V	
输入电压	VI	Vss	Vdd	
工作温度		0 -20 -30	+50 +70 +80	常温型 宽温型 超宽温型
存储温度		-20 -30 -40	+70 +80 +85	常温型 宽温型 超宽温型

(2) 电特性

项目	符号	条件	MIN	TYP	MAX	单位
输入高电压	V _{IH}		2.2	—	—	V
输入低电压	V _{IL}		—	—	0.6	V
输出高电压	V _{OH}	I _{OH} =0.2mA	2.4	—	—	V
输出低电压	V _{OL}	I _{OL} =1.2mA	—	—	0.4	V
电源电流	I _{DD}		—	1.0	2.0	mA


二、接口引脚功能

引脚号	符号	状态	功 能
1	V _{SS}		电源地
2	V _{DD}		+5V 逻辑电源
3	V _O		液晶驱动电源
4	RS	输入	寄存器选择 1: 数据; 0: 指令
5	R/W	输入	读、写操作选择 1: 读; 0: 写
6	E	输入	使能信号
7	DB0	三态	数据总线 (LSB)
8	DB1	三态	数据总线
9	DB2	三态	数据总线
10	DB3	三态	数据总线
11	DB4	三态	数据总线
12	DB5	三态	数据总线
13	DB6	三态	数据总线
14	DB7	三态	数据总线 (MSB)
15	LEDA	输入	背光+5V
16	LEDK	输入	背光地

注：接口顺序不同的模块可能不相同；15、16 两脚用于带背光模块，不带背光的模块这两个脚悬空不接。


三、接口特性及时序

写操作 (MPU 至 HD44780)



项 目	符号	最小值	最大值	单位
使能周期	Tc	1000	—	ns
使能脉冲宽度	Tw	450	—	ns
使能升、降时间	Tr, Tf	—	25	ns
地址建立时间	Tsp1	140	—	ns
地址保持时间	Thd1	10	—	ns
数据建立时间	Tsp2	195	—	ns
数据保持时间	Thd2	10	—	ns

读操作（MPU 至 HD44780）



项 目	符号	最小值	最大值	单位
使能周期	Tc	1000	—	ns
使能脉冲宽度	Tw	450	—	ns
使能升、降时间	Tr, Tf	—	25	ns
地址建立时间	Tsp1	140	—	ns
地址保持时间	Thd1	10	—	ns
数据延迟时间	Td	—	320	ns
数据保持时间	Thd2	10	—	ns

● 功能描述

一，接口

字符型液晶显示模块和主机有两种接口：4 位数据总线和 8 位数据总线。两种总线的选择取决于指令“功能设置”中的“DL”位。使用 4 位数据总线时使用接口中的 DB4~DB7，8 位的数据或指令码分两次写入或读出，先处理高 4 位，然后是底 4 位。

在主机对液晶显示模块进行读写操作时，有两个寄存器被用做数据的暂存。一个是数据寄存器（DR），一个是指令寄存器（IR）。数据寄存器用来临时存储要写入到 DDRAM /CGRAM 中或要从 DDRAM/CGRAM 中读出的数据。要写或读哪个 RAM 取决于之前的地址设置指令。每次读或写数据操作时，RAM 和寄存器间的数据的传输是自动完成的。指令寄存器用来存储从主机传送过来的指令代码。但主机不能读出指令代码。数据寄存器还是指令寄存器的选择真值表如下：

RS	R/W	E	功能
0	0	下降沿	写指令代码（主机写 IR）
0	1	高电平	读忙标志和 AC 码
1	0	下降沿	写数据（主机写 DR）
1	1	高电平	读数据（主机读 DR）

二，忙标志（BF）

BF=“1”时，说明 HD44780 正在进行内部操作，此时对它传送的任何指令都将被忽略。所以当执行指令时请确认 BF 是“0”。BF 可以通过 RS=“0”、R/W=“1”时读出“DB7”得到。

三，地址计数器（AC）

地址计数器用来存储 DDRAM/CGRAM 的地址。当进行一次写或读操作后，地址计数器自动加一或减一。地址计数器可以用读操作读出。

四，显示数据存储（DDRAM）

DDRAM 可以最多存储 80 个字符码。DDRAM 的地址指针存储在 AC 中。DDRAM 中的内容和 LCD 上显示的内容的对应关系如下（在没有进行移位、滚屏操作时）：

1) 8 列 1 行（N=0， F=0）：

字符位	1	2	3	4	5	6	7	8
DDRAM 地址	00H	01H	02H	03H	04H	05H	06H	07H

2) 8 列 2 行（N=1， F=0）：

Line	字符位	1	2	3	4	5	6	7	8
1	DDRAM 地址	00H	01H	02H	03H	04H	05H	06H	07H
Line	字符位	1	2	3	4	5	6	7	8
2	DDRAM 地址	40H	41H	42H	43H	44H	45H	46H	47H

3) 16 列 1 行（N=1， F=0）：

字符位	1	2	7	8	9	10	15	16
DDRAM 地址	00H	01H	06H	07H	40H	41H	46H	47H

4) 16 列 2 行 (N=1, F=0):

Line	字符位	1	2	3	4	15	16
1	DDRAM 地址	00H	01H	02H	03H	0EH	0FH
Line	字符位	1	2	3	4	15	16
2	DDRAM 地址	40H	41H	42H	43H	4EH	4FH

5) 16 列 4 行 (N=1, F=0):

Line	字符位	1	2	3	4	15	16
1	DDRAM 地址	00H	01H	02H	03H	0EH	0FH
Line	字符位	1	2	3	4	15	16
2	DDRAM 地址	40H	41H	42H	43H	4EH	4FH
Line	字符位	1	2	3	4	15	16
3	DDRAM 地址	10H	11H	12H	13H	1EH	1FH
Line	字符位	1	2	3	4	15	16
4	DDRAM 地址	50H	51H	52H	53H	5EH	5FH

6) 20 列 1 行 (N=0, F=0):

字符位	1	2	3	4	19	20
DDRAM 地址	00H	01H	02H	03H	12H	13H

7) 20 列 2 行 (N=1, F=0):

Line 1	字符位	1	2	3	4	19	20
	DDRAM 地址	00H	01H	02H	03H	12H	13H
Line 2	字符位	1	2	3	4	19	20
	DDRAM 地址	40H	41H	42H	43H	52H	53H

8) 20 列 4 行 (N=1, F=0):

Line	字符位	1	2	3	4	19	20
1	DDRAM 地址	00H	01H	02H	03H	12H	13H
Line	字符位	1	2	3	4	19	20
2	DDRAM 地址	40H	41H	42H	43H	52H	53H
Line	字符位	1	2	3	4	19	20
3	DDRAM 地址	14H	15H	16H	17H	26H	27H
Line	字符位	1	2	3	4	19	20
4	DDRAM 地址	54H	55H	56H	57H	66H	67H

9) 24 列 2 行 (N=1, F=0):

Line 1	字符位	1	2	3	4	23	24
	DDRAM 地址	00H	01H	02H	03H	16H	17H
Line 2	字符位	1	2	3	4	23	24
	DDRAM 地址	40H	41H	42H	43H	56H	57H

10) 40 列 2 行 (N=1, F=0):

Line 1	字符位	1	2	3	4	39	40
	DDRAM 地址	00H	01H	02H	03H	26H	27H
Line 2	字符位	1	2	3	4	39	40
	DDRAM 地址	40H	41H	42H	43H	66H	67H

11) 40 列 4 行，此类产品使用两颗 HD44780 控制器（N=1，F=0）：

IC1:

Line 1	字符位	1	2	3	4	39	40
	DDRAM 地址	00H	01H	02H	03H	26H	27H
Line 2	字符位	1	2	3	4	39	40
	DDRAM 地址	40H	41H	42H	43H	66H	67H

IC2:

Line 3	字符位	1	2	3	4	39	40
	DDRAM 地址	00H	01H	02H	03H	26H	27H
Line 4	字符位	1	2	3	4	39	40
	DDRAM 地址	40H	41H	42H	43H	66H	67H

五、字符发生器（CGROM）

字符发生器中存有 204 个 5X8 点阵字符和 32 个 5X11 点阵字符，如下图。

Standard Character Pattern

Upper 4bit Lower 4bit															
	LLLL	LLHL	LLHH	LHLL	LHLH	LHHL	LHHH	HLLL	HLLH	HLHL	HLHH	HHLL	HHLH	HHHL	HHHH
LLLL	CG RAM (1)														
LLH	(2)														
LLHL	(3)														
LLHH	(4)														
LHLL	(5)														
LHLH	(6)														
LHHL	(7)														
LHHH	(8)														
HLLL	(1)														
HLLH	(2)														
HLHL	(3)														
HLHH	(4)														
HHLL	(5)														
HHLH	(6)														
HHHL	(7)														
HHHH	(8)														



Thu Jun 05 08:41:48 1997

注：上图为 HD44780（KS0066U）的标准字库。若使用的液晶显示模块的字库不是标准字库，则请参考具体产品附带的规格书。

六，用户字符发生器（CGRAM）

用户可以利用 CGRAM 制作最多 8 个 5x8 点阵字符。将字模数据写进 CGRAM，就可以象使用 CGROM 一样使用它们。

字符码、CGRAM 地址和字模数据之间的关系如下(阴影部分为“1”):

字符码	CGRAM 地址	CGRAM 中的字模数据	字模号
D7 D6 D5 D4 D3 D2 D1 D0	A5 A4 A3 A2 A1 A0	P7 P6 P5 P4 P3 P2 P1 P0	
0 0 0 0 0 0 0 0	0 0 0 0 0 0 0 0 1 0 1 0 0 1 1 1 0 0 1 0 1 1 1 0 1 1 1	x x x 0  0	pattern 1
*	* * * * * *	*	*
0 0 0 0 0 1 1 1	1 1 1 0 0 0 0 0 1 0 1 0 0 1 1 1 0 0 1 0 1 1 1 0 1 1 1	x x x  0	pattern 8

● 指令描述

1、清屏

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	0	0	0	0	0	0	1

运行时间（250Khz）：1.64ms；

功能：清 DDRAM 和 AC 值。

2、归位

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	0	0	0	0	0	1	*

运行时间（250Khz）：1.64ms；

功能：AC=0，光标、画面回 HOME 位。

3、输入方式设置

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	0	0	0	0	1	I/D	S

运行时间（250Khz）：40us；

功能：设置光标、画面移动方式。

其中：I/D=1：数据读、写操作后，AC 自动增一；

I/D=0：数据读、写操作后，AC 自动减一；

S=1：数据读、写操作，画面平移；

S=0：数据读、写操作，画面不动；

4、显示开关控制

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	0	0	0	1	D	C	B

运行时间（250Khz）：40us；

功能：设置显示、光标及闪烁开、关。

其中：D 表示显示开关：D=1 为开，D=0 为关；

C 表示光标开关：C=1 为开，C=0 为关；

B 表示闪烁开关：B=1 为开，B=0 为关。

5、光标、画面位移

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	0	0	1	S/C	R/L	*	*

运行时间（250Khz）：40us；

功能：光标、画面移动，不影响 DDRAM。

其中：S/C=1：画面平移一个字符位；

S/C=0：光标平移一个字符位；

R/L=1：右移；R/L=0：左移。

6、功能设置

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	0	1	DL	N	F	*	*

运行时间（250Khz）：40us；

功能：工作方式设置（初始化指令）。

其中：DL=1，8 位数据接口；DL=0，四位数据接口；

N=1，两行显示；N=0，一行显示；

F=1，5×10 点阵字符；F=0，5×7 点阵字符。

7、CGRAM 地址设置

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	1	A5	A4	A3	A2	A1	A0

运行时间（250Khz）：40us；

功能：设置 CGRAM 地址。A5~A0=0~3FH。

8、DDRAM 地址设置

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	1	A6	A5	A4	A3	A2	A1	A0

运行时间（250Khz）：40us；

功能：设置 DDRAM 地址。

N=0，一行显示 A6~A0=0~4FH；

N=1，两行显示，首行 A6~A0=00H~27H，

次行 A6~A0=40H~67H。

9、读 BF 及 AC 值

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	1	BF	AC6	AC5	AC4	AC3	AC2	AC1	AC0

功能：读忙 BF 值和地址计数器 AC 值。

其中：BF=1：忙；BF=0：准备好。

此时，AC 值意义为最近一次地址设置（CGRAM 或 DDRAM）定义。

10、写数据

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
1	0	数 据							

运行时间（250Khz）：40us；

功能：根据最近设置的地址性质，数据写入 DDRAM 或 CGRAM 内。

11、读数据

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
1	1	数				据			


运行时间 (250Khz)：40us；

功能：根据最近设置的地址性质，从 DDRAM 或 CGRAM 数据读出。


● 应用指南

一、供电电路和对比度调节

1，液晶驱动电压小于 5V 的产品：




2，液晶驱动电压大于 5V 的产品：



二、直接访问方式接口电路及驱动程序

直接访问方式的接口电路如下图所示：



直接访问方式的驱动子程序如下：

A15=E 的选通信号，A9=R/W，A8=RS

```

COM      EQU      20H      ; 指令寄存器
DAT      EQU      21H      ; 数据寄存器
CW_Add   EQU      8000H    ; 指令口写地址
CR_Add   EQU      8200H    ; 指令口读地址
DW_Add   EQU      8100H    ; 数据口写地址
DR_Add   EQU      8300H    ; 数据口读地址

```

1、读 BF 和 AC 值

```

PRO:      PUSH      DPH
          PUSH      DPL
          PUSH      ACC
          MOV        DPTR, #CR_Add    ; 设置指令口读地址
          MOVX       A, @DPTR         ; 读 BF 和 AC 值
          MOV        COM, A           ; 存入 COM 单元
          POP        ACC
          POP        DPL

```

POP DPH
RET

2、写指令代码子程序

```
PR1:    PUSH    DPH
        PUSH    DPL
        PUSH    ACC
        MOV     DPTR, #CR_Add    ; 设置指令口读地址
PR11:   MOVX     A, @DPTR
        JB      ACC.7, PR11      ; 判 BF=0? 是继续
        MOV     A, COM
        MOV     DPTR, #CW_Add    ; 设置指令口写地址
        MOVX    @DPTR, A        ; 写指令代码
        POP     ACC
        POP     DPL
        POP     DPH
        RET
```

2、写显示数据子程序

```
PR2:    PUSH    DPH
        PUSH    DPL
        PUSH    ACC
        MOV     DPTR, #CR_Add    ; 设置指令口读地址
PR21:   MOVX     A, @DPTR
        JB      ACC.7, PR21      ; 判 BF=0? 是继续
        MOV     A, DAT
        MOV     DPTR, #DW_Add    ; 设置数据口写地址
        MOVX    @DPTR, A        ; 写数据
        POP     ACC
        POP     DPL
        POP     DPH
        RET
```


4、读显示数据子程序

```
PR3:    PUSH    DPH
        PUSH    DPL
        PUSH    ACC
        MOV     DPTR, #CR_Add    ; 设置指令口读地址
PR31:   MOVX     A, @DPTR
        JB      ACC.7, PR31      ; 判 BF=0? 是继续
        MOV     DPTR, #DR_Add    ; 设置数据口读地址
        MOVX    A, @DPTR        ; 读数据
        MOV     DAT, A          ; 存入 DAT 单元
        POP     ACC
```

POP DPL
POP DPH
RET

三、间接控制方式接口电路及驱动程序

间接控制方式（4 位数据接口）是利用 HD44780 所具有的 4 位数据总线的功能，简化电路接口的一种接口方式。实用电路如下：



注：间接控制方式也可以用 8 位数据接口的方式，直接控制也可以使用 4 位数据接口方式，实际应用时可以灵活掌握。4 位数据接口方式适用 4 位单片机或 8 位单片机接口不够用的情况，因 4 位数据传输时每个字节的数据或指令需要需传输两次，所以比较麻烦，建议在单片机接口够用，还是把 8 位数据口都接上。

间接控制方式的驱动子程序如下：

RS	EQU	P3.3	；寄存器选择信号
R/W	EQU	P3.4	；读/写选择信号
E	EQU	P3.5	；使能信号

1、读 BF 和 AC 值

PRO : PUSH ACC

MOV	P1, #0FFH	; P1 置位, 准备读
CLR	RS	; RS=0
SETB	R/W	; R/W=1
SETB	E	; E=1
MOV	COM, P1	; 读 BF 和 AC6-4 值
CLR	E	; E=0
MOV	P1, #0FFH	; P1 置位, 准备读
SETB	E	; E=1
MOV	A, P1	; 读 AC3-0 值
CLR	E	; E=0
SWAP	A	; 转换成 8 位数据
ANL	A, #0FH	
ANL	COM, #0F0H	
ORL	A, COM	
MOV	COM, A	; 送入 COM 单元
POP	ACC	
RET		

2、写指令代码子程序

PR1 :	PUSH	ACC	
	CLR	RS	; RS=0
	SETB	R/W	; R/W=1
PR11:	MOV	P1, #0FFH	; P1 置位, 准备读
	SETB	E	; E=1
	MOV	A, P1	; 读 BF 和 AC6-4 值
	CLR	E	; E=0
	MOV	C, ACC. 7	BF->进位位 C
	SETB	E	; E=1
	CLR	E	; E=0 (读 AC3-0 值)
	JC	PR11	; 判别 BF=1? 是转
	CLR	R/W	; R/W=0
	MOV	P1, COM	; 写入指令代码高 4 位
	SETB	E	; E=1
	CLR	E	; E=0
	MOV	A, COM	; 写入指令代码低四位
	SWAP	A	
	MOV	P1, A	
	SETB	E	; E=1
	CLR	E	; E=0
	POP	ACC	
	RET		

3、写显示数据子程序

PR2 :	PUSH	ACC
-------	------	-----

	CLR	RS	; RS=0
	SETB	R/W	; R/W=1
PR21:	MOV	P1, #0FFH	; P1 置位, 准备读
	SETB	E	; E=1
	MOV	A, P1	; 读 BF 和 AC6-4 值
	CLR	E	; E=0
	MOV	C, ACC. 7	BF->进位位 C
	SETB	E	; E=1
	CLR	E	; E=0 (读 AC3-0 值)
	JC	PR21	; 判别 BF=1? 是转
	SETB	RS	; RS=1
	CLR	R/W	; R/W=0
	MOV	P1, DAT	; 写入数据高 4 位
	SETB	E	; E=1
	CLR	E	; E=0
	MOV	A, DAT	; 写入数据低 4 位
	SWAP	A	
	MOV	P1, A	
	SETB	E	; E=1
	CLR	E	; E=0
	POP	ACC	
	RET		

4 、读显示数据子程序

PR3 :	PUSH	ACC	
	CLR	RS	; RS=0
	SETB	R/W	; R/W=1
PR31:	MOV	P1, #0FFH	; P1 置位, 准备读
	SETB	E	; E=1
	MOV	A, P1	; 读 BF 和 AC6-4 值
	CLR	E	; E=0
	MOV	C, ACC. 7	; BF->进位位 C
	SETB	E	; E=1
	CLR	E	; E=0 (读 AC3-0 值)
	JC	PR31	; 判别 BF=1? 是转
	SETB	RS	; RS=1
	SETB	R/W	; R/W=1
	MOV	P1, #0FFH	; P1 置位, 准备读
	SETB	E	; E=1
	MOV	COM, P1	; 读数据高 4 位
	CLR	E	; E=0
	MOV	P1, #0FFH	; P1 置位, 准备读
	SETB	E	; E=1
	MOV	A, P1	; 读数据低 4 位


```

CLR      E          ; E=0
SWAP     A          ; 转换成 8 位
ANL      A, #0FH
ANL      DAT, #0F0H
ORL      A, DAT
MOV      DAT, A      ; 数据送入 DAT 单元
POP      ACC
RET

```

四、应用程序

该程序是根据液晶显示模块 GTC-40021 而编制的。其他规格的字符型液晶显示模块都适用，仅是地址的改动而已。

示例一 初始化子程序

(1) 直接访问方式下的初始化子程序

```

INT      MOV      A, #30H          ;
          MOV      DPTR, #CW_Add   ; 指令口地址设置
          MOV      R2, #03H        ; 循环量=3
INT1:    MOVX     @DPTR, A          ; 写指令代码
          LCALL    DELAY            ; 调延时子程序
          DJNZ     R2, INT1
          MOV      COM, #38H        ; 设置工作方式*
          LCALL    PR1              ;
          MOV      COM, #01H        ; 清屏
          LCALL    PR1
          MOV      COM, #06H        ; 设置输入方式
          LCALL    PR1
          MOV      COM, #0FH        ; 设置显示方式
          LCALL    PR1
          RET

```

(2) 间接控制方式下的初始化子程序

```

INT      MOV      P1, #30H          ; 工作方式设置指令代码
          CLR      RS              ; RS=0
          CLR      R/W             ; RW=0
          MOV      R2, #03H        ; 循环量=3
INT1     SETB     E                ; E=1
          CLR      E                ; E=0
          LCALL    DELAY            ; 调延时子程序
          DJNZ     R2, INT1
          MOV      P1, #28H        ; 设置工作方式*
          SETB     E                ; E=1
          CLR      E                ; E=0

```

```

MOV      COM, #01H      ; 清屏
LCALL    PR1
MOV      COM, #06H      ; 设置输入方式
LCALL    PR1
MOV      COM, #0FH      ; 设置显示方式
LCALL    PR1
RET

```

注：*工作方式参数根据所使用的字符型液晶显示模块的特性而定。

```

DELAY:   MOV      R6, #00H      ; 延时子程序
          MOV      R7, #00H
DELAY1:   NOP
          DJNZ     R7, DELAY1
          DJNZ     R6, DELAY1
          RET

```

初始化演示程序

```

MAIN:    MOV      SP, #60H
          ANL      P3, #0C7H
          LCALL    INT

```

示例二 字符的写入方式

HD44780 有五种字符写入方式。演示程序如下：

(1) 逐字依次输入方式演示程序段

```

          MOV      COM, #06H
          LCALL    PR1
          MOV      COM, #80H
          LCALL    PR1
          MOV      DPTR, #TAB
          MOV      R2, #12
          MOV      R3, #00H
WRIN:     MOV      A, R3
          MOVC     A, @A+DPTR
          MOV      DTA, A
          LCALL    PR2
          LCALL    DELAY
          INC      R3
          DINZ     R2, WRIN
          SJMP     $
TAB:      DB      43H, 75H, 72H, 73H, 6FH, 72H, 20H      ; Cursor
          DB      57F, 72H, 69H, 74H, 65H                ; write

```

(2) 光标左移输入方式演示程序段

```

CL_ENTER: MOV      COM, #04H      ; 输入方式设置
          LCALL    PR1
          MOV      COM, #90H      ; 设置 DDRAM 地址

```

```

                LCALL    PR1
                MOV      DPTR, #TABCL          ; 设置字表首地址
                MOV      R2, #12H              ; 循环量设置
CL_1:          MOV      A, R2
                DEC      A
                MOVC     A, @A+DPTR            ; 取字符代码
                MOV      DAT, A
                LCALL    PR2                    ; 写数据
                LCALL    DELAY                  ; 演示用延时
                DJNZ     R2, CL_1
                SJMP     $
TABCL:         DB   43H, 75H, 72H, 73H, 6FH, 72H, 20H      ; Cursor
                DB   4CH, 65H, 66H, 74H, 20H              ; Left
                DB   53H, 63H, 72H, 6FH, 6CH, 6CH          ; Scroll

```

(3) 光标右移输入方式演示程序段

```

CR_ENTER:      MOV      COM, #06H              ; 输入方式设置
                LCALL    PR1
                MOV      COM, #80H              ; 设置 DDRAM 地址
                LCALL    PR1
                MOV      DPTR, #TABCR          ; 设置字表首地址
                MOV      R2, #13H              ; 循环量设置
                MOV      R3, #00H
CL_1:          MOV      A, R3
                MOVC     A, @A+DPTR            ; 取字符代码
                MOV      DAT, A
                LCALL    PR2                    ; 写数据
                INC      R3
                LCALL    DELAY                  ; 演示用延时
                DJNZ     R2, CR_1
                SJMP     $
TABCR:         DB   43H, 75H, 72H, 73H, 6FH, 72H, 20H      ; Cursor
                DB   52H, 69H, 67H, 68H, 74H, 20H          ; Right
                DB   53H, 63H, 72H, 6FH, 6CH, 6CH          ; Scroll

```

(4) 画面左滚动输入方式演示程序段

```

L_ENTER:       MOV      COM, #07H              ; 输入方式设置
                LCALL    PR1
                MOV      COM, #90H              ; 设置 DDRAM 地址
                LCALL    PR1
                MOV      DPTR, #TABL           ; 设置字表首地址
                MOV      R2, #0BH              ; 循环量设置
                MOV      R3, #00H
L_1:           MOV      A, R3
                MOVC     A, @A+DPTR            ; 取字符代码

```

```

        MOV     DAT, A
        LCALL   PR2                ; 写数据
        INC     R3
        LCALL   DELAY              ; 演示用延时
        DJNZ    R2, L_1
        SJMP    $
TABL:    DB     4CH, 65H, 66H, 74H, 20H        ; Left Scroll
        DB     53H, 63H, 72H, 6FH, 6CH, 6CH
(5) 画面右滚动输入方式演示程序段
R_ENTER: MOV     COM, #05H          ; 输入方式设置
        LCALL   PR1
        MOV     COM, #90H          ; 设置 DDRAM 地址
        LCALL   PR1
        MOV     DPTR, #TABR        ; 设置字表首地址
        MOV     R2, #0CH           ; 循环量设置
R_1      MOV     A, R2
        DEC     A
        MOVC    A, @A+DPTR         ; 取字符代码
        MOV     DAT, A
        LCALL   PR2                ; 写数据
        LCALL   DELAY              ; 演示用延时
        DJNZ    R2, R_1
        SJMP    $
TABR:    DB     52H, 69H, 67H, 68H, 74H, 20H        ; Right Scroll
        DB     53H, 63H, 72H, 6FH, 6CH, 6CH

```

示例三 建立自定义字符库

字符型液晶显示模块允许用户在 HD44780 片内 CGRAM 内（64 个字节 00H-3FH）自建立 8 个 5×8 点阵的字符。建立自定义字符库的子程序如下：

```

CG_WRITE: MOV     COM, #40H          ; 设置 CGRAM 地址
        LCALL   PR1
        MOV     R2, #64             ; 循环量设置
        MOV     R3, #00H
        MOV     DPTR, #CGTAB        ; 设置字模数据首地址
CG1:      MOV     A, R3
        MOVC    A, @A+DPTR
        MOV     DAT, A
        LCALL   PR2
        INC     R3
        DJNZ    R2, CG1
        RET
CGTAB:    DB     08H, 0FH, 12H, 0FH, 0AH, 1FH, 02H, 02H        ; “年” 代码=00H
        DB     0FH, 09H, 0FH, 09H, 0FH, 09H, 11H, 00H        ; “月” 代码=01H
        DB     1FH, 11H, 11H, 1FH, 11H, 11H, 1FH, 00H        ; “日” 代码=02H

```

```

DB  11H, 0AH, 04H, 1FH, 04H, 1FH, 04H, 00H    ; “$” 代码=03H
DB  0EH, 00H, 1FH, 0AH, 0AH, 0AH, 13H, 00H    ; “元” 代码=04H
DB  18H, 18H, 07H, 08H, 08H, 08H, 07H, 00H    ; “℃” 代码=05H
DB  04H, 0AH, 15H, 04H, 04H, 04H, 04H, 00H    ; “↑” 代码=06H
DB  17H, 15H, 15H, 15H, 15H, 15H, 17H, 00H    ; “10” 代码=07H

```

自定义字符显示演示程序段

```

                LCALL    CG_WRITE                ; 调建立自定义字符子程序
                MOV      COM, #80H                ; 设置 DDRAM 地址
                LCALL    PR1
                MOV      DPTR, #TABDY             ; 设置字表首地址
                MOV      R2, #18                  ; 循环量设置
                MOV      R3, #00H
LOOP1:          MOV      A, R3
                MOVC     A, @A+DPTR               ; 取字符代码
                MOV      DAT, A
                LCALL    PR2                      ; 写数据
                INC      R3
                DZJ      R2, LOOP1
                MOV      COM, #0C0H               ; 设置 DDRAM 地址
                LCALL    PR1
                MOV      R2, #18                  ; 循环量设置
LOOP2:          MOV      A, R3
                MOVC     A, @A+DPTR               ; 取字符代码
                MOV      DAT, A
                LCALL    PR2                      ; 写数据
                INC      R3
                DZJ      R2, LOOP2
                SJMP     $
TABDY:          DB  4DH, 44H, 4CH, 53H, 34H, 30H, 32H, 36H, 36H
                DB  20H, 03H, 32H, 35H, 35H, 2EH, 30H, 30H, 04H
                DB  31H, 39H, 39H, 37H, 00H, 37H, 01H, 31H, 02H
                DB  20H, 54H, 3DH, 33H, 35H, 05H, 06H, 20H, 07H

```

冀诚公司销售网络:

大陆工厂

地址: 河北省石家庄市新石北路 368 号 邮政编码: 050091

电话: +86-311-3856940 传真: +86-311-3851891

电子邮件: sales@gemtech-hb.com

深圳宾康实业有限公司

地址: 深圳福田区华强北路宝华大厦 A1205 室

邮政编码: 518031 电话: +86-755-83742919 传真: +86-755-83742632

冀诚电子杭州分公司

地址: 杭州市文苑路 522 号 4-2-601

电话: +86-571-88988297

香港代理

宾康有限公司

地址: 香港九龙湾宏照道 11 号 宝隆中心 B 座 3 字楼 12 室

电话: +852-27518691 传真: +852-27965670

电子邮件: pantage@asiansources.com

新加坡代理

大欣工业私人有限公司

地址: 41 Kallang Pudding Road #06-06 Golden Wheel Building Singapore 349316s

电话: +65-7463198 传真: +65-7468348

电子邮件: tassinsingnet.com.sg
